

NULLWERTE IN ERGEBNISZELLEN VERMEIDEN MIT WENN UND ODER

Häufig entstehen in Ergebniszellen aus mathematischen Gründen Nullen, welche vielleicht „unschön“ aussehen, wenn in der Tabelle noch keine Werte gepflegt wurden. Genau diese fehlenden Vorgänger sind für die Null-Werte verantwortlich. Mit Hilfe der WENN-Funktion, verschachtelter WENN-Funktionen oder der Verschachtelung der WENN- mit der ODER-Funktion kann man die Anzeige unnötiger Null-Werte unterbinden. Es gibt darüber hinaus noch weitere Möglichkeiten.

WENN-FUNKTION

Syntax: =WENN(Prüfung; Dann_Wert; Sonst_Wert)

	A	B	C	D	E	F	G	H	I
1									
2		Netto	Steuersatz	Steuerbetrag					
3					←	=WENN(C3="";"";B3*C3)			
4		500,00 €			←	=WENN(C4="";"";B4*C4)			
5			0,19	0,00 €	←	=WENN(C5="";"";B5*C5)			
6		500,00 €	0,19	95,00 €	←	=WENN(C6="";"";B6*C6)			
7									

Abbildung 1

Die einfache WENN-Funktion in Spalte D überprüft mit dem ersten Argument, ob die jeweilige Zelle in Spalte C leer ist. Diese Prüfung kann als Ergebnis nur WAHR oder FALSCH ergeben, also 1 oder 0.

Hinweis: Der Zustand „Leer“ wird in Formeln durch zwei aufeinanderfolgende Anführungszeichen ausgedrückt. Hier darf kein Leerzeichen eingefügt werden, denn eine Zelle mit Leerzeichen ist nicht leer, sondern enthält ein Zeichen – nämlich das Leerzeichen.

Ergibt die Prüfung WAHR, d.h. die Zelle ist leer, kommt der Dann_Wert zum Zuge. In Abbildung 1 bedeutet das, wenn die Zelle mit dem Steuersatz leer ist, dann lasse auch die Ergebniszelle leer.

Ergibt die Prüfung FALSCH, d.h. die Zelle ist nicht leer, kommt der Sonst_Wert zum Zuge. In Abbildung 1 bedeutet das, wenn die Zelle mit dem Steuersatz nicht leer ist, dann führe die Berechnung aus, die in der Formel als Sonst_Wert eingetragen ist.

Dies ist die Grundüberlegung, die aber noch nicht das optimale Ergebnis zeigt, wie man in Zeile 5 sieht. Um das Ergebnis zu optimieren, gibt es verschiedene Lösungen. Weiter unten erkläre ich beispielhaft die Kombination WENN- und ODER-Funktion.

Zunächst jedoch kurz zur ODER-Funktion.

ODER- FUNKTION

Syntax: =ODER(Wahrheitswert1; [Wahrheitswert 2]; ...)

	A	B	C	D	E	F
1						
2		Zahl 1	Zahl 2	Oder-Funktion	Ergebnis als Zahl	
3				WAHR	1	
4		10		WAHR	1	
5			12	WAHR	1	
6		10	12	FALSCH	0	
7						

Abbildung 2

Formel in D3: =ODER(B3="";C3="")

Formel in E3: =D3*1

In Abbildung 2 wird in Spalte D in jeder Zeile geprüft, ob die jeweilige Zelle in Spalte B oder in Spalte C leer ist. Diese Prüfung kann wieder nur WAHR oder FALSCH ergeben. In Spalte E ist das Ergebnis jeweils durch Multiplikation mit der Zahl 1 als Zahl dargestellt. Wie oben schon beschrieben, kann man mit Wahrheitswerten rechnen: WAHR = 1 und FALSCH = 0. In unserem Beispiel werden nur zwei Wahrheitswerte abgeprüft, nämlich der Zustand der Zelle mit der Zahl 1 und der Zustand der Zelle mit der Zahl 2. Daraus ergeben sich für die Tabelle in Abbildung 2 folgende Ergebnisse:

Prüfen, ob Zelle leer	Prüfen, ob Zelle leer	ergibt zusammen	als Zahl
WAHR	WAHR	WAHR	1
FALSCH	WAHR	WAHR	1
WAHR	FALSCH	WAHR	1
FALSCH	FALSCH	FALSCH	0

Tabelle 1

D.h., wenn eine der beiden oder beide Zellen leer sind, ergibt die ODER-Funktion das Ergebnis WAHR.

WENN UND ODER VERSCHACHTELT

Syntax in diesem Beispiel: =WENN(ODER(Wahrheitswert1; Wahrheitswert2); Dann_Wert; Sonst_Wert)

	A	B	C	D	E	F	G	H	I
1									
2		Netto	Steuersatz	Steuerbetrag					
3					← =WENN(ODER(B3="";C3="");"";B3*C3)				
4		500,00 €			← =WENN(ODER(B4="";C4="");"";B4*C4)				
5			0,19		← =WENN(ODER(B5="";C5="");"";B5*C5)				
6		500,00 €	0,19	95,00 €	← =WENN(ODER(B6="";C6="");"";B6*C6)				
7									

Abbildung 3

Jetzt kombinieren wir die WENN-Funktion mit der ODER-Funktion. Während wir im ersten Beispiel getestet haben, ob eine der beiden Vorgängerzellen leer ist, nämlich der Steuersatz, prüfen wir nun, ob eine oder beide Vorgängerzellen leer sind. Dazu ersetzen wir in der WENN-Funktion der Abbildung 1 die einfache Prüfung durch eine ODER-Funktion.

Dadurch wird geprüft, ob die Zelle „Netto“ oder die Zelle „Steuersatz“ leer ist. Ist dies der Fall, bleibt auch die Ergebniszelle leer (Dann_Wert). Wenn wir der Beschreibung der ODER-Funktion folgen, bleibt die Ergebniszelle auch leer, wenn beide Vorgängerzellen leer sind.

Enthalten beide Vorgängerzellen Werte, wird in der Ergebniszelle das Ergebnis der als Sonst_Wert angegebenen Formel berechnet.